
COMP 3002 Winter 2021 Assignment #1 

Getting Used to Smalltalk 



First lecture Monday Jan 11 6pm, Assignment due date Tuesday Midnight Jan 19.

Note that the questions essentially goes through the 
SMALLTALK CHEAT SHEET in the order given 
(so you might want to use it as a reference).

Hand-in only the code for the last question in a text file. Put it in a folder with 
your name and ID on it. Ignore little errors when you find them.

	Starting Up

0. Assuming you are running on a Windows machine, the first thing to do after you unzip the Smaltalk folder into one of your folders such as a folder called “3002.2021” is to read the file “SETUP INSTRUCTIONS.txt”. It tells you how to setup the system’s PATH variable to refer to Smalltalk. If you did it correctly, Smalltalk should start up when you double click on “VDEVW.EXE”. If it starts to come up but hangs there, the only thing I can think of that could cause that is that your default language for your machine is not English.

1. Use the Windows key + I keyboard shortcut to open the Settings app. (On my computer, the 4 keys on the bottom left are CTRL, Fn, the Windows key, and ALT). 
2. Click Time & language.
3. Click Language.
4. Under Languages, make English the temporary default while you use Smalltalk.

	Working in a workspace and fixing variables you accidentally made global

1.	Create a workspace to work in by selecting “Smalltalk\New Workspace”. The advantage of a workspace is that you do not have to declare your variables. Try it by executing

			test := 1

    To execute it, you must type it in, select it, right click it to bring up a pop-up menu and select either “Do it”, “Show it”, or “Inspect it”. The first executes it without showing anything. The second executes it and prints the result (which is selected allowing you to easily delete it by clicking on the delete key). The third brings up an inspector allowing you to additionally look inside the object.

	Now try the same thing in the Transcript window (the light blue-green window). You will get an error message

			“Define 'test' as global variable?”

    Click on NO. You will then get ‘undeclared’ in front of it. Note that it is selected allowing you to easily delete it by clicking on the delete key. To do this without an error message in the Transcript window, you will need to write the following instead

		| test | test := 1

	Now try it. Most programming languages require your variables to be declared including Smalltalk (but Smalltalk doesn’t want to know what type of variable it is). 

	Do not ever click ‘YES’ to the query “Define 'test' as global variable?” because that is a mistake which you will have to fix. Try executing “test := 1” again in the Transcript window and deliberately click on ‘YES’ to the error message. Here’s what you have to do to fix it.

   Type “Smalltalk’, select it (or double click it which is faster than selecting it), and click “Inspect it’. You will get an inspector on the SystemDictionary. This is a list of all the global variables in Smalltalk. The system dictionary has two columns: the column on the left contains global variable names and the column on the right shows its value.

	Scroll down until you find the global variable name “Smalltalk”. It should have “SmallInteger” just above it. When you click on SmallInteger (on the left), you will see the value “SmallInteger” (on the right). If you double click on “SmallInteger” (on the left), you will get on inspector showing you that “SmallInteger” is a class… You can tell it’s a class because the title at the top of the inspector is 

		“Inspecting: SmallInteger class”

	and what you see are the variables inside the SmallInteger class. Close the inspector and double click on “Smalltalk” (on the left). Yes you get another inspector on the System Dictionary. Just close the second inspector and go back to the original.

	This time, search for ‘test’ in the inspector. Once you find it, single click on it to verify that it’s value is 1. Bring up the pop-up menu above the selected ‘test’ variable. Select the last entry which should be “Remove”. The variable ‘test’ that you accidentally made global is then removed.

	Using inspectors to tell what something is!

2. Open an inspector on each one of the following to find out what it is (the title at the top of the inspector tells you what it is). You can double click on any field to further look inside it. 

		nil  true  false  'hi'   #hi  $h  10 10.2   #(10 20)

	Distinguishing equality from identity

3.	Try to predict the results of the following 6 tests before you execute it…

			'hi' = 'hi'     'hi' == 'hi'                 	#hi = #hi    #hi == #hi             6 = 6      6 == 6

	Executing NOTHING and executing a COMMENT

4. Try excuting just spaces and executing a comment like "This is a comment."

	Dot versus Semicolon

5. Dot is a statement terminator and a statement separator. For example, the following 2 if statements have the assignment “b := 20” terminated with dot (left example) or not (right example).

		a := b := 0.					a := b := 0.
		4 > 3 ifTrue: [a := 10. b := 20.]			4 > 3 ifTrue: [a := 10. b := 20]

	The right example is the preferred way of writing the code. The semicolon, by contrast with other languages, does NOT terminate a statement. It indicates that “you wish to send another message to the same receiver that received a message earlier” WITHOUT REPEATING THE RECEIVER. For example, instead of writing
			aCollection := OrderedCollection new.
			aCollection add: 10.
			aCollection add: 20.
			aCollection add: 30.
			^aCollection "This returns the collection"

	which works, you can write	

			aCollection := OrderedCollection new.
			aCollection add: 10; add: 20; add: 30.
			^aCollection "This returns the collection"

	Note that the second and third “add:” don’t repeat the receiver. By the way, what happens if we misspelled OrderedCollection and wrote OrderCollection instead. If you don’t know, try it now.

	Here’s an interesting question? What does the “add:” method return. For example, try

			aCollection := OrderedCollection new.
			^aCollection add: 10
	
	Surely, there can only be 2 possibilities. It either returns the collection or it returns 10. Verify that it returns 10.

	Consider the method “yourself”. What do you get if you execute

			10 yourself			#hi yourself			Array yourself

	That’s right… You get the object you sent the message to… So we can now replace the 3 statements

			aCollection := OrderedCollection new.
			aCollection add: 10; add: 20; add: 30.
			^aCollection "This returns the collection"

	by one statement as follows:
	
		^OrderedCollection new add: 10; add: 20; add: 30; yourself

	Check it out. This is the preferred way of building a long collection. We can do the same with dictionaries as follows:

		^Dictionary new at: #red put: #rouge; at: #green put: #vert; yourself

	If you just print the result, you won’t be able to tell if it’s actually correct but you will if you inspect it.

	If you actually understood all this, then you should be able to predict the result of the following

		^1+2; * 3

	Hint: it’s either 9 or 3?

	Outputting into the Transcript

6.	Just like C++ can output into cout for debugging, Smalltalk can output into the Transcript for debugging. Try

		Transcript << 'Hi there'

    You can then restore the Transcript window with “CNTL r” (r for RESTORE). You can use semicolon to output more than one thing.

		Transcript << 'The sum of 1 and 3 is '; << (1+3).

	By the way, you need the brackets around 1+3. Try it without the brackets.

	How do we output onto a new line? The short of it is we need to look it up… Not on the internet but in the Smalltalk system itself. But first, what kind of object is Transcript. Try printing or inspecting

		Transcript

	Once you know what it is, open a browser by clicking on “Smalltalk\Browse classes” which can also be done by clicking on “CNTL b” (b for browse). Once the browser opens up, right button click on the top left pane and click on “Find class…”. Then type in the class of whatever “Transcript” was. Once the correct class appears, you will see a button with “instance” in it. Click on it, and it will switch to “class”. Click on it again and it will switch to “instance”. The Transcript is an instance of this class, so we need to look for a method that will output a new line on the instance side.

	If you click on “extensions”, you will see that “<<” is a method there (1 of 6 methods). Nothing seems to have anything to do with outputting a new line. So try clicking on “no category methods”. Nothing there either.

	Above “TranscriptWindow” is “TextWindow” (the fact that it’s above means TranscriptWindow inherits from TextWindow). It won’t take you long to find “cr” which indicates that it creates a “line-feed”, an archaic way of saying a new line. So let’s try it.

		Transcript cr; << 'Did it work?'

	Type conversion

7.	The conventional way of converting an object is to think “I want you AS something else”. For example,

		10 asFloat
		10.2 asInteger
		10.5 asString
		true asString
		'hello' asSymbol
		#hi asString

     The most general way to convert an object is to ask it for its printString but we can also ask for its storeString.

		anObject printString
		anObject storeString

	For example,

		(OrderedCollection new add: 10; add: 20; add: 30; yourself) printString
				gives you 'OrderedCollection (10 20 30)'

	whereas

		(OrderedCollection new add: 10; add: 20; add: 30; yourself) storeString
				gives you '((OrderedCollection new) add: 10; add: 20; add: 30; yourself)'

	So the storeString creates the code that will create an object equal to the receiver.

	In contrast, if you create a class that you want to have the ability to print or store itself, you need to add a method called

		printOn: aStream
		storeOn: aStream

	where the stream behaves just like the transcript. You don’t need to create one because that’s done by printString or storeString which is already implemented.

	To give you an example of how we might implement method “printOn:” for a boolean, we could implement the method as follows:

		printOn: aStream
			"Convert the boolean to a string."
			aStream << (self ifTrue: ['true'] ifFalse: ['false'])

	Note that “self” (like “this” in other languages) is the boolean. Output into the stream is done with an if-expression rather than an if-statement because it’s slightly shorter. Using an if-statement would look like the following:

		printOn: aStream
			"Convert the boolean to a string."
			self ifTrue: [aStream << 'true'] ifFalse: [aStream << 'false']

	Simple blocks

8.	Blocks in Smalltalk (using square brackets) are the equivalent of scopes (using braces) in other languages. So instead of writing

			C++							Smalltalk
			if (a < b) {x = 1;} else {x = 2;}			a < b ifTrue: [x := 1] ifFalse: [x := 2]

			while (a < b) {a = a + 1;}				[a < b] whileTrue: [a := a + 1]

			for (long x = 0; x < 10; x++) {a = a + x;}		0 to: 9 do: [:x | a = a + x]

	For the most part, we went from braces to square brackets. But blocks are a little bit more than scopes, they are in fact functions with no names which allows them to be executed over and over again. 

	If you think of ifTrue:ifFalse: as a message, you can see that the receiver is a boolean (the result of executing a < b). So it’s implemented in the boolean class. 

	By contrast, whileTrue: as a message has a receiver which is a block. Since whileTrue: has to be able to execute a < b many times, it’s actually a function with no name. Hence whileTrue: is implemented in the class Block. So the whileTrue: in Smalltalk is slightly different from the while in C++. 

	Finally, to:do: as a message has a receiver which is the integer 0. So it’s implemented in the Integer class. It’s its job to provide the block with a variable x that keeps changing, which is why x is a block variable.
 
	That’s probably enough about blocks for now lest you get a headache. But keep in mind that blocks are objects. So it allows us to do the following EVEN THOUGH YOU MIGHT NOT BE ABLE TO THINK OF A USE FOR THAT FEATURE.
		
		b1:= [x := 1].	b2 := [x := 2].
		a < b ifTrue: b1 ifFalse: b2

		b3 := [a < b]. b4 := [a := a + 1].
		b3 whileTrue: b4

		b5 := [:x | a = a + x].
		0 to: 9 do: b5

	If this code is inside a method, the blocks might have been passed in as parameters and it would all work.

	Collections

9.	Although there are 30 different kinds of collections in Smalltalk, only 3 are super useful: arrays, ordered collections, and dictionaries containing key-value pairs (with 2 variations), and once in a while, sorted collections. All of them except sorted collections can contain mixtures of different kinds of objects. When indexable by integers, indexing starts at 1 rather than 0.

		Making one						Initializing one		Accessing one

		x := Array new: 10					x at: 1 put: 10			x at: 1
		x := OrderedCollection new				x add: 10			x at: 1 	
		x := Dictionary new					x at: #John put: 10		x at: #John
		x := SortedCollection sortBlock: [:x :y | x <= y]		x add: 10			x at: 1 	

	Arrays are fixed size but the others can grow without limits. They all understand size. So “x size” will return 10 for the array but 1 for all of the others. As I said, anything can be put into a collection.

			For the array, 			x at: 2 put: 'hello'
			For the ordered collection, 	x add: 'hello'
			For the dictionary, 		x at: true put: 'hello'
			For sorted collections,		it must be possible to compare the elements (so you can’t mix types)
	
	For sorted collections, the sort block just needs to return true if the 2 parameters are already sorted. The way it works is that it compares 2 elements at a time, essentially saying “Are these 2 sorted, no? so I’ll flip them”
	“Are these 2 sorted, yes? So I’ll leave them alone”, etc.

	"example sort block 1"	[:x :y | x <= y] 		"value true if x <= y leads to increasing values"
	"example sort block 2"	[:x :y | x >= y] 		"value true if x >= y leads to decreasing value"

	Note that when a key-value pair is provided to a dictionary using at:put:,  if the key doesn’t match anything that already exists, a new key-value pair is added. But if it key already matches, it replaces the old value by the new value.

		Matching based on = is called a Dictionary (the ordinary kind)
		Matching based on == is called an IdentityDictionary

	Let’s verify the dictinction between the two types of dictionaries. First, try the following.

 		object1 := 'hi'.
 		object2 := object1 copy.
 		object3 := object1 copy.
		dictionary := Dictionary new.
		dictionary at: object1 put: #Object1.
		dictionary at: object2 put: #Object2.
		dictionary at: object3 put: #Object3.
		dictionary at: object1. "This should give you #Object3"
		dictionary at: object2. "This too should give you #Object3"

	Then replace Dictionary by IdentityDictionary and try again. Now, the values you get should be #Object1 and #Object2. We’ll find out midterm how useful identity dictionaries can be when trying to make a copy of a finite state machine (whatever that is).

	It’s useful to be able to create arrays and ordered collections with pre-initialized elements using what we refer to as the with:with: notation. For example,

		Array with: 10
		Array with: 10 with: 20
		Array with: 10 with: 20 with: 30
				…
		OrderedCollection with: 10
		OrderedCollection with: 10 with: 20
		OrderedCollection with: 10 with: 20 with: 30
				…

	You can’t do that with dictionaries though. Also, there are a host of short forms for accessing indexable collections like arrays and ordered collections. Try

		test := OrderedCollection with: 10 with: 20 with: 30.

	Then instead of saying test at: 1, test at: 2, and test at: 3, try

		test first		test second		test third
		test last		test secondLast	test thirdLast
		test reverse

	Querying Indexable Collections

10. There are a bunch of routines you can use to determine if something is in a collection.
	aCollection includes: anObject		"true if there using ="
	aCollection includesIdentical: anObject	"true if there using =="
	indexOf: anObject				Which one if equal one there. 0 if missing"
	indexOf: anObject ifAbsent: [nil] 		"Like above but nil of not there."
	indexOfIdentical: anObject			Which one if identical one there. 0 if missing"
	indexOfIdentical: anObject ifAbsent: [nil] 	"Like above but nil of not there."

	One version is looking for a specific object (the version with the word “identical”). Another is looking for an object equal to it (the version without “identical). Let’s use an array literal for trying out these searches. As always, try to guess the answers before you try it.

			#(10 20 'hi' 'hi' #(big barn) @W) includes: 10
			#(10 20 'hi' 'hi' #(big barn) @W) includes: 'hi'
			#(10 20 'hi' 'hi' #(big barn) @W) includesIdentical: 'hi'
			#(10 20 'hi' 'hi' #(big barn) @W) includes: #(big barn) 
			#(10 20 'hi' 'hi' #(big barn) @W) includesIdentical: @W

	Did you figure out that there is only ONE uppercase W in Smalltalk? Since Smalltalk is 1-based rather than 0-based like C++ or Java, the index routines on one of the queries above should be some number from 1 to 6 if it’s there and 0 if it’s not. Try

			#(10 20 'hi' 'hi' #(big barn) @W) indexOf: @W
			#(10 20 'hi' 'hi' #(big barn) @W) indexOf: 30aCollection add: anObject			"Adding at the end"
aCollection addLast: anObject		"Adding at the end"
aCollection addFirst: anObject		"Adding at the start"
aCollection addIfAbsent: anObject		"Adding if not there using ="
aCollection addIfIdenticalAbsent: anObject	"Adding if not there  using =="
aCollection addAllIfAbsent: anCollection	"Adding many if not there using ="
aCollection addIfIdenticalAbsent: anObject	"Adding many if not there using =="


	If you don’t like getting back 0 when it’s  not there, you can use the version that allows a different result if its not there

			#(10 20 'hi' 'hi' #(big barn) @W) indexOfIdentical: 50 ifAbsent: [nil]
			#(10 20 'hi' 'hi' #(big barn) @W) indexOfIdentical: 50 ifAbsent: [-100+1]

	You can compute anything you want in the “ifAbsent:” block.

	OrderedCollection add routines

11. There are a bunch of routines for adding things into an ordered collection, including

	anOrderedCollection add: anObject				"Adding at the end"
	anOrderedCollection addLast: anObject			"Adding at the end"
	anOrderedCollection addFirst: anObject			"Adding at the start"
	anOrderedCollection addIfAbsent: anObject			"Adding if not there using ="
	anOrderedCollection addIfIdenticalAbsent: anObject	"Adding if not there  using =="
	anOrderedCollection addAllIfAbsent: anCollection		"Adding many if not there using ="
	anOrderedCollection addAllIfIdenticalAbsent: anObject	"Adding many if not there using =="
	
	You can use an ordered collection as a stack using addLast and removeLast (not shown in the list), or as a queue using addLast and removeFirst (also not shown). More typically, we just use “add:” but sometimes we want to make sure it’s not added twice so we use “addIfAbsent:” or “addIfIdenticalAbsent:”. Note that

		OrderedCollection new addIfAbsent: #(10 20 10 10)

	will end up with an ordered collection with the array #(10 20 10 10) inside because we told it to add the array. But if we just want the CONTENTS of the array to be added (not the array itself), you could use

		OrderedCollection new addAllIfAbsent: #(10 20 10 10)

	Try them both out to make sure you get it…

	Dictionary routines

12. We’ve already seen how to create a dictionary (or an identity dictionary) earlier, namely

	Dictionary new
		at: key1 put: object1; 
		at: key2 put: object2;
		… 
		at: keyn put: objectn;
		yourself

	where we had to use “yourself” because the last “at:put:” returns “objectn” rather than the dictionary. There are a few obvious routines for manipulating dictionaries, namely

	aDictionary at: key1 			"Accessing. Error if not there"
	aDictionary at: key1 ifAbsent: [nil]	"nil if not there"
	aDictionary at: key1 put: object1	"Changing"
	aDictionary includes: value
	aDictionary includesKey: key
	aDictionary keys asArray
	aDictionary values asArray

	Normally, we created the dictionary so we know what’s in it but if we’re not sure, we can use “at:ifAbsent:” just like “indexAt:ifAbsent:” for ordered collections or arrays. If we just want to find out if it contains a value or a key without retrieving it, we can use “includes:” (for a value) and “includesKey:” for a key. If we want all the keys or all the values, we can uses “keys” or “values”. Notice the plural form of the message.  Now, you might wonder why I added “asArray” at the end? We’ll it’s because get 2 non-standard kinds of collection that you are not allowed to index. But they’re easy to convert. But first, do a little experiment to find out what kinds of collection you get.

	Using Browsers

13. To illustrate a simple Temperature class with celsius degrees, if we want to use temperature objects as keys in a dictionary, we will need to ensure it supports 2 specific methods. They are easy to implement. But first, how do we create a class. Unfortunately, there is NO CREATE CLASS button. In this question, we will examine the use of browsers and leave the creation of a new class for the next question.

	Start by opening a browser by clicking on pulldown menu “Smalltalk\Browse classes” or “CNTRL b” (b for browse) and scroll down to the bottom until you see “Random Experimentation”. Then go up a bit to “3D-Basic”. Then click on “BoundingBox”. Then, click on the INSTANCE or CLASS button a few times until you leave it with INSTANCE visible. Finally, click on “get\set” and then “center”. You now have something in each pane.

	The four panes at the top contain

		“class categories” 	“class names”		“method categories”	  “method names”

	and the bottom pane is the text for the method called “center”. What the browser is saying is that there are 4 classes in the class category “3D-Basic”, namely BoundingBox, Point3D, Point4D (a subclass of Point3D; notice that it’s indented), and WorldGlobal.

	Similarly, “get/set” is one of 8 method categories containing 8 methods: “center”, “center:”, “maximum”, etc. 

	The classes and methods are permanent (in the class names pane and method names pane, the second and fourth panes). The class categories pane is just for organizing classes together that are related; it’s possible to add or rename the class categories without changing anything. Similarly, the method categories pane is just for organizing methods together that are related; it’s possible to change or rename the method categories without changing anything.

	Now go to the method text pane and while the mouse is above that pane, right click your mouse. This is the pop-up menu for that pane. Since there are 5 panes, there are 5 different pop-up menus. Try them out just to see the differences.

	Now go to the method text pane and type a space at the end of a period just so you know you changed something. The text goes red. Now type “CNTRL z” which is undo. The text goes back to black. Try it again but this time, instead of typing undo, get the pop-up menu and select “Restore” (it’s the same as undo). Make it go red again but this type, select “Save” from the pop-up method. You have just compiled your first method.

	Now go to the class names pane, bring up the pop-up menu, and click on “Definition”. You should see

		Object subclass: #BoundingBox
  		instanceVariableNames: 
			' maximum minimum center radius '
  		classVariableNames: ''
 	 	poolDictionaries: ''
  		categories: #( '3D-Basic' )
  		comment: ''

	This is code to create the class BoundingBox. The receiver is Object (the class Object). The message has 6 keywords (count them). 

	The “subclass:” keyword is followed by the symbol for the name of the class to create (it starts with # to indicate that it’s a symbol). 

	The “instanceVariableNames:” keyword is a string containing the names of the instance variables. Currently, the BoundingBox class has 4 instance variables. Instance variables are only visible to instances. You could edit this by renaming some variables, adding more, removing others. 

	There are no class variables. If there were, they would be visible to both instances and the class.

	There are no pool dictionaries. For our course, we’ll never need them anyway.

	The class category keyword is indicated by an array literal containing the string '3D-Basic'. If we wanted to put BoundingBox in a second category, we would change the array literal to something like 

		#('3D-Basic' 'My own category')

	and now, BoundingBox would be in 2 different categories. Finally, the “comment:” keyword allows you to add a string describing this class. As you can see, the designer of this class (me) didn’t bother with it. 

	Before we try to change this class to see what happens, let’s just make an instance somewhere in a workspace. Here’s two things to try. Write the following code, select it, and inspect it so we can see inside the object.

		BoundingBox new
		BoundingBox unitBox

	You will find that the instance variables of the first one are all nil but the instance variables of the second one are all initialized. By the way, you can you find the “unitBox” method in BoundingBox. Hint: it’s a class method (not an instance method).

	Ok, now we’re ready for an experiment. We’re going to redefine this class but first make a copy of the definition (for example, by copying the definition to the Transcript) so you can undo it afterward. Here a new definition for this class:

		Object subclass: #BoundingBox
  		instanceVariableNames: 
			' maximum minimum center radius junk1 junk2'
  		classVariableNames: ''
 	 	poolDictionaries: ''
  		categories: #( '3D-Basic'  'Gaming classes')
  		comment: 'A box that can indicate the bounds of an object.'

	You just select the code and click on “Do it”. This will add 2 instance variables, add a new class category and change the comment. After you’ve done it, verify that junk1 and junk2 now exists by inspecting

		BoundingBox new 
		
	To find the new class category “Gaming classes”, you will have to scroll down in the class category pane until you see it. If it’s not there, it’s because the browser doesn’t know the compiler added it. Just open a new browser. You’ll find it there at the bottom. If you click on it, you will see that it contains the BoundingBox class that “3D-Basic” contained but only that one. We just added category “Gaming classes” to BoundingBox, not Point3D or Point4D.

	Now that we added 2 instance variables, it’s customary to add a get method and a set method for each one. You do that by clicking on “Get and Set…” in the class name pane. A get and set builder tool with pop up. The idea is to associated a type with each new variable. Before you start, notice the pane containing 

			“** New Name **”. 

	It’s an editable pane. Select junk1 on the left and select a type on the right, like Integer. Then select junk2 and select a different type on the right, say String. Notice that it’s a finite list. Suppose you wanted the name Kangaroo which is NOT in the list. Then you can change your mind and change String to type Kangaroo by typing Kangaroo in the editable pane. 

	One other issue. Many instance variables are perfectly fine initialized with nil since other code you will write will typically set them up. But one thing that must be specifically set up are variables initialized to collections; e.g., specific size arrays, empty ordered collections, empty dictionaries, etc. That typically is done by a method called initialize. There’s a toggle button at the bottom of the get/set builder 'New/Initialize' which you can click on to turn on. Let me repeat. You don’t toggle it on all the time. Only when collection variables are needed. For this situation, DO NOT TURN ON THE TOGGLE. If you did, just untoggled it. 

	So lastly you finish off by clicking on BUILD. Actually do it.

	Here what will have happened after you did all that. There will now be a get method for junk1 which will return the junk1 variable and a set method junk1: which will allow you to change the junk1 variable.

		Use of a variable junk1 looks like:   		junk1 := 10.			Left side of an assignment
									otherVariable := junk1 + 1	Right side of an assignment
		Use of the junk1 method looks like: 		aBoundingBox junk1		It has a receiver
		Use of the junk1: method looks like: 		aBoundingBox junk1: 100	It has a receiver too

	You can’t mix up get and set methods from a variable. Get and set methods have a receiver. Variables DO NOT have a receiver.

	You will also have a get and a set method for junk2. Lastly, IF YOU HAD TOGGLED ON the 'New/Initialize' button, it would have created 2 methods and if this class already had these 2 methods, it would replace the old versions by the new one. We normally do that when we create the class the first time with all of the instance variables it is meant to have. What are the 2 methods

		new		On the class side. It invokes the new method of the superclass to get an uninitialized 					instance and then sends the initialize message to this unintialized object which it then 
					returns.

		initialize		On the instance side which attempt to initialize and return the instance.

	The new method is always written correctly but not the initialize method. It attempts to write the initialize method correctly but it often does things that are wrong because it’s not as smart as a human. So it’s up to you to check the method just in case. An example mistake I’ve seen it do is initialize a boolean, say called isTall, using code like

		isTall := Boolean new
	
	Surely, you can’t do this in any programming language. It should have said either

		isTall := true.

	Or

		isTall := false.

	Ok… Now undo all that; i.e., redefine it the way it had already been defined initially. One thing it will not do is remove the 4 get and set methods for junk1 and junk2. Use the CTRL key to multiple select the 4 methods in the method names pane, get the pop-up menu for pane, and click on “Remove” to delete those methods. If you look at the bottom of your Transcript, you will notice that the compiler gave an error message for each of these 4 methods because they are referring to instance variables junk1 and junk2 that no longer exist.

	Similarly, the 'Gaming classes' category will not be removed just in case it’s in use by other classes. Go find it and use the “class categories” pop-up menu to remove it.

	One last question. How many browsers do you currently have on your screen.. More than one is too many. Close them all. If you don’t adopt a strategy for using browsers, it won’t take long before you have too many and this will slow your programming down. 

	Recommentation: Try to have just one active browser on the screen and if you need to go look for something, rather than use this browser to do that, open a second one, go look for what you want, and then when you find it, close that second browser. Right underneath it will be the first browser ready for you to continue with. Note: it will take you a while to work like this. You’ll know you forgot to do this because every now and then, you will notice that you have dozens of browsers open.

	Ok… We’re ready to create a new Temperature class.

	Creating a simple class

14. We’re going to implement a simple Temperature class and we’re going to put it in the 'Random Experimentation' class category. Start by scrolling down to the bottom of the class categories pane (the top left one). There you will find 'Random Experimentation'. Select it. That should show you which classes are in that class category and they should appear in the classes name pane (the second one from the top). It’s empty (there are no classes in that category) Get the pop-up menu for that pane and click on Definition. The following should appear in the text pane (the bottom on).

			SuperClass subclass: #NameOfNewClass
 				instanceVariableNames: ''
  				classVariableNames: ''
 				poolDictionaries: ''
  				categories: #( 'Random Experimentation' )

	This is sample code that you need to fill in. Every class needs a superclass. The top of the class hierarchy is Object. So replace Superclass by Object (all classes are capitalized, so don’t type ‘object’). Next replace NameOfNewClass by Temperature (that’s our new class). Don’t delete #. What instance variables do we need. We just need one. Let’s call it centigrade, so type that in between the single quotes. We won’t need class variables and we won’t ever need pool dictionaries, so leave them as they are (you can’t delete those keywords because all keywords must be kept). Finally, it typed the category we want. So you don’t need to change anything. If we wanted to additionally put it in the 'Table Constructor' category which is just above 'Random Experimention', we would add it to the array literal #( 'Random Experimentation' ). But we don’t need to do that. If you did all that, it should look like 

			Object subclass: #Temperature
 				instanceVariableNames: 'centigrade'
  				classVariableNames: ''
 				poolDictionaries: ''
  				categories: #( 'Random Experimentation' )

	Keep in mind that this is a piece of code. You are sending a message to the class Object to create a subclass called Temperature. But there is no class called Temperature yet (what you provide instead was the symbol #Temperature, which is a unique string) and you told it that it’s instances (when you make one) will have one instance variable called 'centigrade' and that this class should be placed in the class category 'Random Experimentation’

	All that’s left to do is compile it. Can you remember how? YES, bring up the pop-up menu in the pane containing this code and click on “Save”.

	To see that it worked, go into a workspace and inspect all three of the following

		#Temperature
		Temperature
		Temperature new

	The first is a symbol, the second is the Temperature class, and the third is an instance of Temperature. You can see that the instance has an instance variable class called “centigrade”. Close your inspectors and carry on. 

	Now, create the get and set methods for centigrade indicating that it’s type is a Float. I’m going to let you re-figure out how to do this since you’ve done it before. Beside, you have detailed instructions for doing this in the previous question if you need to look. After you’re done, you will have a “get/set” method category and two instance methods. Take a look at them.

	Try inspecting an initialized instance by executing

		Temperature new centigrade: 0.0

	That was simple. This class is done. What the heck, let’s make another addition. I would like to create one more method to ensure that it legal to execute the following code:

		Temperature centigrade: 0.0

	But it’s not legal yet. Try it. You’ll get a red debugger appearing with the error message ‘centigrade: not understood’ in the title bar. 

	Also, 4 lines down you see the line “Temperature class (Object)>>doesNotUnderstand:”. It’s saying “the Temperature class which inherits from class Object does not understand the message centigrade:”. That’s because “centigrade:” is an instance method, not a class method.

	Click on Debug and the debugger will open up, scroll down to the 4th line, and click on “self” and then on “message”. From “self”, you can tell that the receiver is the class but “message” is too big for the screen. So double-click on message instead. Look at “receiver”, “selector”, then “arguments”. It’s telling you that the class Temperatures received a message with selector “centigrade:” with arguments containing 0.0. And that’s illegal because the class DOES NOT HAVE A METHOD called centigrade:. So close the debugger and let’s check out the methods that are understood by the class.

	Click on the INSTANCE button in the browser. It switches to CLASS which means you now see the class methods. All you see is “no category methods” and if you click on it, there is nothing inside. By the way, if you hold the CTRL key down while you click on “no category methods”, it will deselect.

	Let’s create a new method category before we create a new method. Get the pop-up menu for the method category pane (the third from the top) and click on Add..  Then type in “instance creation”. All method categories are in lowercase. Not only is the method category selected but also the text in the bottom pane is “('instance creation' )”. Let’s just overwrite it complexly by a new method

		create: aFloat
			^self new centigrade: aFloat

	This is creating the CLASS METHOD called create that takes one parameter aFloat. The method refers to self (which has to be the class Temperature since we are in a class method for Temperature). Saying “self new” is equivalent to saying “Temperature new” and we get an instance back. Then we send the instance (not the class) the message centigrade: to modify its instance variable call “centigrade” and then it returns whatever “centigrade:” returns. If you go look at the “centigrade:” method, you will see that there is NO “^” in the method which essentially means the code ran until the end of the method. At the end of every method is an implied “^self” and since self in the method called “centigrade” is an instance (not the class), we get the instance back.

	After having created this method “create:”, the following code would be legal. Try it by inspect the result of executing

		Temperature create: 10.0

	Now go back to the create method and modify it by changing “create:” to “createAgain:” and save it. This is to show you that if you modify an existing method name, it does not modify that method but instead creates a new one with the new name.

	Allright, go back to the create: method again and change the name to “centigrade:’ and save it. Now for the consequences. There are now two methods called “centigrade:”, one on the CLASS side and one on the INSTANCE side. Notice that they do 2 different things.	If you now inspect the result of executing

		Temperature centigrade: 20.0

	It works. Sending “centigrade:” to the class causes the “centigrade:” method in the class to execute which creates an instance using “new” and then sends “centigrade:” to the instance causing it to execute the instance method “centigrade:” which modifies the “centigrade” instance variable and returns that modified instance. Which goes to show that we don’t need method “create” or method “createAgain”.

	Remove them both at the same time using multiple-select. You should be able to figure it out how by yourself.

	Now for the purpose of this section. I want to be able to use Temperature objects as keys in a Dictionary. For example, I want to be able to do the following. To make it easier to test this out more than once, create a new category on the class side called examples and add a method called example1 defined as shown below. Note that when an existing category is selected, adding a new category goes above it. If you deselect all categories, adding a new category goes after existing categories.

		example1
			"Temperature example1"
			| dictionary |
			dictionary := Dictionary new
			at: (Temperature centigrade: 30.0) put: #Hi;
			at: (Temperature centigrade: 20.0) put: #Medium;
			at: (Temperature centigrade: 0.0) put: #Low;
				yourself.
			^dictionary at: (Temperature centigrade: 30.0) 

	Note that as you save the above method, you get a caution with “You are redefining a superclass method?” All this means is that the superclass must also already have a method called example1. Since that is true and you also want an example1 method in Temperatures, just click on OK. Lots of classes have there own example1 method. 

	The expectation is that when we run “Temperature example1” inside the comment, we get back #Hi. Do we? Try it out. You can see that you get a debugger appear with the message “key is missing”? Just close it. Now let’s consider fixing it…

	Problem 1: It doesn’t know how to compare 2 temperature objects. Comparing is done with the operator =. So we need to add a method for that. First let me do it not quite correctly in the hopes that you can figure out what’s going to be wrong before I tell you. Then we’ll fix it. Now, is it to be an instance method or a class method. An instance method: It’s two instances we want to compare, not two classes. So create a method category for that, say called 'comparing' and add the following method.

		= anObject
			^self centigrade = anObject centigrade

	Note that a method like the above is used like the following,

		(Temperature centigrade: 5.0) = (Temperature centigrade: 5.0) 

	The receiver, the first “(Temperature centigrade: 5.0)” is referred to as self in the method and the parameter, the second “(Temperature centigrade: 5.0)” is referred to as “anObject” in the method. If you run the code above, you should get back 

		true
	
	because the centigrade instance variables in the two different temperature objects are equal. So it works but there is still a bug. Can’t figure it out. We’ll then, try

		(Temperature centigrade: 5.0) = 'hello'

	Should this bring up a debugger or should it just say false (it’s not equal). It’s the latter. We can easily fix it by making use of the message class. For example,

		'hello' class					should give you back String
		(Temperature centigrade: 5.0) class		should give you back Temperature

	So fix the equal method as follows:

		= anObject
			anObject class = Temperature ifFalse: [^false].
			^self centigrade = anObject centigrade

	All we are saying is that if anObject is not a temperature object, then self which has to be a Temperature object because it’s the receiver of the “=” message can’t possibly be equal to it. So return false. Without this, “anObject centigrade” would have generated an error message when anObject was the string 'hello' because you can’t ask a string for it’s centigrade amount. So now

		(Temperature centigrade: 5.0) = 'hello'		should give you back false

	It does yet example1 still doesn’t work.
	
	Problem 2: Any object that can be used as a key must be hashable; i.e., must have a method called “hash” that returns an integer. It doesn’t matter what the value actually is. What matters is that IF TWO OBJECTS ARE EQUAL, THEY MUST HAVE the same hash.

	For fun, execute the following to see that they all have a hash and what those hashes are…

		20210000000 hash
		'hello' hash
		2021.0 hash
		#hello hash
		#(10 20 30) hash
		(Temperature centigrade: 5.0) hash
		(Temperature centigrade: 5.0) hash

	Apparently, the hash works for temperature objects. It must be inheriting hash from Object, the superclass. But unfortunately, every hash in Object is unique. You get 10 different hashes even if the 10 objects look the same… You need to make sure that if the two objects are equal, their hashes are the same. You will note that if you take the hash of 'hello' twice, for example, you always get the same hash. The trick is to compute a hash for YOUR object out of objects that are known to be hashable. Since you object contains an instance variable “centigrade” that contains a float, you just need to use it’s hash as your hash. If it had more than one instance variable, you could add or multiply or subtract the two hashes. So try adding a simple method like

		hash
			^centigrade hash

	It’s returning the hash of you centigrade variable OR you could have written hash using the get method

		hash
			^self centigrade hash

	Verify that example1 now works before you start working on the next question.

	Looping Routines that DON’T return a value

15. There are many ways of looping over a collection. The “do:” routine (see below is the most used).

	aCollection do: [:index | …]					Loops over values
	aDictionary keysDo: [:key | ..]					Loops over keys 
	aDictionary keysAndValuesDo: [:key :value | ..]		Loops over key-value pairs 
	#((1 2) (3 4) …) groupsDo: [:value1 :value2 | ..]		Loops over pairs 
	#((1 2 3) (4 5 6) …) groupsDo: [:value1 :value2 :value3 | ..]	Loops over triples 
	#((1 2 3) (4 5 6) …) reverseGroupsDo: [:value1 :value2 :value3 | ..]	Loops in the backwards order
	#(a b …) indexedDo: [:index :object | ...]			Incrementing index number from 1 onward
	#(a b …) reverseIndexedDo: [:index :object | ...]		Decrementing index number from 
									last index going forward
	#(a b ..) with: #(1 2 …) do: [:element1 :element2 …]		Gets a with 1, b with 2, etc. Make
									sure the 2 collections are the same size

	I’d like you to try each one of the above out to find out how they work. To give you an idea, I’ll use “do:” on both an array and a dictionary because it works on all kinds of collections. Also, so you can see what it actually does, I’ll output something into the transcript… Trying it out on an array, you might try.

		#(10 20 30) do: [:object | Transcript cr; << 'Object is '; << object].

	and you should get the following output in the transcript.

		Object is 10
		Object is 20
		Object is 30

	Remember, to clear the transcript before the next experiment, hit “CNTL r” for reset. Trying out

		(Dictionary new at: 10 put: #hi; at: 20 put: #bye; at: 30 put: #ok; yourself) do: [:object | 
			Transcript cr; << 'Object is '; << object].

	you get

		Object is ok
		Object is hi
		Object is bye

	which indicates that “do:” allows you to sequence through the values (not the keys). Of all these sequencing operations, the most useful ones are “do:”, “indexedDo:”, and “keysAndValuesDo:”.
	
	Looping routines that DO return a value

16. The following looping routines are one of the heaviest used features of Smalltalk, in the order collect:, select: (sometimes reject:) and once in while partition: (or its synonym separate:). 

	Keep in mind that you choose the loop variable names. It doesn’t matter what you call them. The two most important collection routines are collect: and select:.

	#(1 2 3) collect: [:object | object * object] => #(1 4 9 …)
	#((1 1) (2 2) …] groupsCollect: [:x :y | x+y] => #(2 4 …]
	#(1 2 3) select: [:object | object odd] => #(1 3…)
	#(1 2 3) reject: [:object | object odd] => #(2 4…)
	#(1 2 3) partition: [:value | value odd] => aDictionary with all odd 
		values associated with true and all false associated with false
	#(1 2 3) separate: [:value | value odd] => a synonym for partition
	#(1 2 3 'hi') separate: [:value | value class] => aDictionary with all integers
		 associated with class SmallInteger and all strings associated 
		with class String
	withoutEquals				All equal objects removed
	withoutIdenticals			All identical objects removed

	Methods collect:, select:, and reject apply to collections like arrays and ordered collections but not to dictionaries. They all build collections that are the same type as the receiver.

	For collect:, each element is used to compute a new value so the resulting collection has the same size as the receiver. Let’s try using it 3 times. Feel free to copy it from this document to Smalltalk to avoid having to type it in. Note: comma is the concatenation operator.

		#(10 20 30) collect: [:integer | integer * 2]
		#('hello' 'hi' 'hey') collect: [:string | string, ' there']
		#(#test 20 30 'hello' 15.1) collect: [:object | object class]

	For select: and reject:, the idea is to create a subset of the existing objects. If you want to specify the ones you want, use select:. If you want to specify the ones you don’t want, use reject:. For example,

		#(1 2 3 4 5 6) select: [:integer | integer even]
		#(1 2 3 4 5 6) reject: [:integer | integer odd]
		#(#test 20 30 'hello' 15.1) select: [:object | object class name size > 6]

	Method groupsCollect: is similar to collect: but has the ability to handle multiple inputs. Write the code that allows 3 integers such as (1 4 0) to convert to the string '1:4-0'.

		#((1 3 6) (3 5 2) (4 5 1) groupsCollect: [YOU WRITE THE CODE]

	Routine partition: (or its synonym separate:) are very powerful routines to partition a collection into subsets. You compute a value that can be used to categorize an object and it uses that category as a key and places all objects together associated with that category. It’s best to try it out.
	
	For example, 
	#(1 2 3) partition: [:value | value true] 
	
		Computes two values: true or false
		So it returns a dictionary with the following key-value pairs

			true 			#(2)
			false			#(1 3)

	Similarly, 
		#(1 2 3 'hi' 20.3 24.4) partition: [:value | value class] 
		
		Computes three values: SmallInteger, String, and Float
		So it returns a dictionary with the following key-value pairs

			SmallInteger 		#(1 2 3)
			String			#('hi')
			Float			#(20.3 24.4)

	I guess we’re going to have to wait until you need it before you find a good use.

	The last two examples are pretty obvious. For example,

		#(1 2 2 3 3 3 4 4 4 4 5 5 5 5 5) withoutEquals is #(1 2 3 4 5). It’s an alternative to using method 
	
	addIfAbsent: (remember that)?

	Decomposing (looks like a loop)

17. The following is a facility that you will only find in this version of Smalltalk and Swift, Apple’s newest programming language. Let me introduce the facility by first reminding you of something you already know. Suppose you had a collection of name-age pairs and you wanted to loop over those pairs to do something. First, let me write the code as if you don’t remember the best way to do it. If that’s the case, try to guess the better way of doing it.

		test
			#((#Jim 21) (#Liu 19) (#Nan 22)) do: [:anObject |
				"Now I need to pick out the pieces. To do that, I need to declare two variables…"

	Starting again, I might rewrite it as  

		test
			| name age |
			#((#Jim 21) (#Liu 19) (#Nan 22)) do: [:pair |
				"Now I can pick out the pieces."
				name := pair first. age := pair second.
				Transcript cr; << 'Processing '; << name; << ' who is '; << age; << ' years old.']

	It’s cumbersome to have had to declare the two variables “name” and “age”. What’s a better way? Try

		test 
			#((#Jim 21) (#Liu 19) (#Nan 22)) groupsDo: [:name :age |
				Transcript cr; << 'Processing '; << name; << ' who is '; << age; << ' years old.']

	By using “groupsDo:” instead of “do:”, I can take advantage of the fact that it allows me to declare 2 loop variables that give me access to the two pieces without having to extract it. This is way nicer. Suppose instead, we don’t have a collection to loop over but instead we have a variable that we know contains a pair, how do we do the same sort of thing.

		test: pair
			| name age |
			name := pair first. age := pair second.
			Transcript cr; << 'Processing '; << name; << ' who is '; << age; << ' years old.'
	Up above, I did it the clumsy way. But since we don’t have a collection to loop over, we can’t use “groupsDo:”. To deal with this, I invented a new facility about a dozen years ago. It turns out that Swift came up with a similar idea. I’ll show you later how I did it (by that I mean, I’ll show you how I implemented it; for now, we’ll just use it).

	I invented a variation of “groupsDo:” which I called “groupDo:” which executes once without a loop. I can use it as follows:

		test: pair
			pair groupDo: [:name :age |
			Transcript cr; << 'Processing '; << name; << ' who is '; << age; << ' years old.']

	What “groupDo:” does is decompose “pair” into two pieces and give me the ability to choose the names of the two pieces so that I don’t have to use assignments statements to do it. I chose “name” and “age”. But I could have called it anything; e.g.
					
		test: pair
			pair groupDo: [:part1 :part2 |
			Transcript cr; << 'Processing '; << part1; << ' who is '; << part2; << ' years old.']

	The “groupDo:” method has no limit to the number of parts I might need to decompose. So it is quite a general method. For example, if there were 4 parts, I would need to decompose it with 4 variables. For example,

		test: fourPartObject
			fourPartObject groupDo: [:name :age :weight :height |
				Transcript cr; << 'Processing '; << name; << ' who is '; << age; << ' years old'.]
				Transcript << ' weighing '; << weight; << ' with height '; << height; << '.']

	Try to avoid manually extracting the pieces out of pairs, triples, quadruples, quintuples, etc in the future.

	Blocks

18. A block is an unnamed function (it’s a feature that has been in Smalltalk since 1972 and was just recently introduced into C++ and Java where they are called lambdas. You can create them, assign them, pass them around as parameters, and execute them. But before we get to those details, here’s a typical example.

	test
		| sum | sum := 0.
		aCollection do: [:x | 
			x < 0 ifTrue: [^-1]. "returns from method test"
			sum := sum + x].  "returns from the block to keep looping"
		^sum

This test method just adds the integers in “aCollection” provided that they are all positive. But if any of the integers are less than or equal to 0, it returns 0. It’s not a very useful method but it is useful to explain a small number of things. 

First notice that 

	[:x | 
		x < 0 ifTrue: [^-1]. 
		sum := sum + x].  

is a one parameter block (the parameter is x). Let me call it the “do:” block. This block is PHYSICALLY passed as a parameter to method “do:” and it’s the job of “do:” to cause what’s inside the block to execute. Basically, if there are 3 integers in “aCollection”, say 10, 20, 30, it is method “do:” that will execute the block 3 times, once when x is 10, once when x is 20, and once when x is 30. The are 3 DIFFERENT places where execution runs. Let me show you the steps as we process

	(1) In method test	(2) In method do: 	(3) Inside the “do:” block code

	      Set sum to 0	

		Calls the block with 10	
	
				Executes code in the block with x = 10.
				Sum becomes 0+10; i.e., 10

		Comes back from the call
		Calls the block with 20

				Executes code in the block with x = 20.
				Sum becomes 10+20; i.e., 30

		Comes back from the call
		Calls the block with 30

				Executes code in the block with x = 30.
				Sum becomes 10+20; i.e., 30

		Comes back from the call
		Decode there are no more to do.
		Returns from method do:

	Returns from method
	test with sum; i.e. 30

Other than the fact that the block is an unnamed function, if you can believe that unnamed functions can be executed, this should make sense even though I’m not showing you the code for method “do:”. 

	It works a little differently if the three integers were -10, 20, 30, however. 

	(1) In method test	(2) In method do: 	(3) Inside the “do:” block code

	      Set sum to 0	

		Calls the block with -10	
	
				Executes code in the block with x = -10.
				Since x < 0, it executes “^-1”.
				This is what’s different. Instead of
				returning from here back to (2), we instead 
				return -1 from (1) DIRECTLY; i.e., from method 
				test.

IT NEVER GETS TO CONSIDERING 20 AND 30 IN THE COLLECTION because once you return from method test, you have returned from the whole thing and you can’t come back to keep looping.

Before I show you the code for method “do:”, let’s consider how you can manipulate blocks. Keep in mind, a block is an object.

	You create blocks by executing something surrounded by square brackets as follows:

	ZeroParameterBlock	:= [1+2]
	OneParameterBlock	:= [:integer | integer + 1]
	TwoParameterBlock	:= [:number1 :number2| : number1 + number2]
	ThreeParameterBlock	:= [:a :b :c | a + b + c]

	Note that the first assignment above does NOT give you 3. It gives you a block that WILL COMPUTE 3 when you execute the block.

	Try inspecting “[]”, a block with no parameters. You will find that is actually calls itself a block closure BUT it’s not able to show you the code that’s inside. Try inspecting “[:integer | integer + 1]”. It’s the same. The best it can do is let you know it’s a block (by the way, nobody actually calls it a block closure).

	To execute a block, you send it either a “value” message, a “value:” message, a “value:value:” message, etc as shown below DEPENDING ON THE NUMBER OF PARAMETERS. The value that comes back after executing the block is shown to the right of “=>”.

	ZeroParameterBlock value					=>	3
	OneParameterBlock value: 10				=>	11
	TwoParameterBlock value: 10 value: 20			=>	30
	ThreeParameterBlock value: 1 value: 2 value: 3		=>	6
		ThreeParameterBlock valueWithArguments: #(1 2 3) 	Same as above

		aBlock argumentCount 					To find out how many parameters
						
	Are you ready to see how method “do:” works? Keep in mind that we are looking at the “do:” method for Array. That means “self” is an array and “self size” will be the size of the array and “self at: x” will be the xth element of the array.

		do: aBlock	
			1 to: self size do: [:index | aBlock value: (self at: index)]

	That’s it. This method knows that aBlock is a 1-parameter block. So it goes through the elements one by one and gives the element to the block using message “value:”. Provided the code gets to the end of the block, the last value computed will come back as the answer (that’s what we call a block return) BUT IT DOESN’T DO ANYTHING WITH IT. Once ”do:” is done, it just returns. Recall, whenever you fall off the end of a method, nil is automatically returned. 

	Also, if a method return is encountered during the execution of the block, the value in the return will return from the method the block is in. It will never get back to this “do:”.

	How hard would it be to implement “collect:” for arrays? Not that much harder. Here is one way of writing it. So that you have a reference point to compare how collect: is used, recalled that you tried examples like

		#(1 2 3 4) collect: [:integer | integer * 10]    and you get back #(10 20 30 40)

	Here’s the method

		collect: aBlock
			| result |
			result := OrderedCollection new.	
			1 to: self size do: [:index | 
				result add: (aBlock value: (self at: index))].
			^result as Array

	It’s a little more complicated but not much more. This method is supposed to return an array but it’s easier to work with an ordered collection since it’s easy to use “add:” to put things  in it. Then we can convert the ordered collection back into an array at the end.

	The idea is simple. Get hold of an element of the collection, execute the block by sending it a value: message with the element as a parameter. When you get the answer back, put it into the ordered collection. When you’re done, return an array instead of an ordered collection (since the receiver was an array, not an ordered collect). The original collection is NOT CHANGED.

	What’s the implication of all this. I’ve just shown you how to implement “do:” and “select:”. If it wasn’t already in Smalltalk, it would have been easy to add. Since “do:” and “select:” are examples of control structures, that means that Smalltalk gives you the ability to add new control structures. NO OTHER LANGUAGE IN THE WORLD ALLOWS YOU TO DO THIS in such a general way. Although the recent addition of lambdas is an attempt to do this, you will find the syntax extremely complex and you won’t like it. Moreover to work properly, you need both the notions of block returns and method returns. C++ and Java provide only block returns. This means it’s very difficult to return from a method when you are inside a lambda. And this is something you do all the time.

	Last example: Here the implementation of “groupDo:” in collection. 

	groupDo: aBlock
		aBlock evaluateWithArguments: self as Array

  	It doesn’t need to loop. It just needs to execute the block once. Routine evaluateWithArguments: can work with an arbitrary number of parameters but it can’t work with ordered collections or dictionaries (only arrays). So convert “self” to an array before attempting to execute the block. 

	If you want to see how other control structures are implemented, just type the name of the control structure possibly with wild cards in a workspace, get the pop-up menu, and click on implementers. If more than one class implements it, you’ll get a list of possibilities. Pick one you’re interested in. For example, try

		indexed*        OR indexedDo:

	I added method “indexedDo:” to this Smalltalk about 10 years ago, well after Smalltalk was invented in 1972. This is very handy for implementing complex software of any type. Any time something is complicated or cumbersome to do, you can add new control structures that make your code simpler. That doesn’t happen in other programming languages.



	Final Task

19. How about putting everything you learned together and handing in just this question. It won’t be complicated. Create a class Truck with the following example class method that will run. Anything this needs to make it work, you will have to add.

		example1
			"Truck example1"
			| aTruck |
			aTruck := Truck new driver: #Jim.
			aTruck addPassenger: #Tom; addPassenger: #Dayton.
			aTruck addLoad: #Wheelbarrow; addLoad: #Ladder; addLoad: #Cement.
			aTruck driverDo: [:driver |
				Transcript cr; << 'The driver is '; << driver].
			aTruck passengersDo: [:passenger |
				Transcript cr; << 'One passenger is '; << passenger].
			aTruck loadDo: [:load |
				Transcript cr; << 'The back of the truck contains a '; << load].
			"Method do: sequences over everything in the truck."
			aTruck do: [:anObject |
				Transcript cr; << 'In the truck, there is a '; << anObject].
	

